Chapter 34
IRCHART Statement

Chapter Table of Contents

Chapter 34
IRCHART Statement

Overview

The IRCHART statement creates control charts for individual measurements and moving ranges. These charts are appropriate when only one measurement is available for each subgroup sample and when the measurements are independently and normally distributed.

You can use options in the IRCHART statement to

- compute control limits from the data based on a multiple of the standard error of the individual measurements and moving ranges or as probability limits
- tabulate individual measurements, moving ranges, and control limits
- save control limits in an output data set
- save individual measurements and moving ranges in an output data set
- read preestablished control limits from a data set
- apply tests for special causes (also known as runs tests and Western Electric rules)
- specify a known (standard) process mean and standard deviation for computing control limits
- specify the number of consecutive measurements to use when computing the moving ranges
- display distinct sets of control limits for data from successive time phases
- add block legends and symbol markers to reveal stratification in process data
- superimpose stars at points to represent related multivariate factors
- clip extreme points to make the chart more readable
- display vertical and horizontal reference lines
- control axis values and labels
- control layout and appearance of the chart

Getting Started

This section introduces the IRCHART statement with simple examples that illustrate commonly used options. Complete syntax for the IRCHART statement is presented in the "Syntax" section on page 1153, and advanced examples are given in the "Examples" section on page 1175.

Creating Individual Measurements and Moving Range Charts

See SHWIR1 in the SAS/QC Sample Library

An aeronautics company manufacturing jet engines measures the inner diameter of the forward face of each engine (in centimeters). The following statements create a SAS data set that contains the diameter measurements for 20 engines:

```
data jets;
   input engine diam @@;
   label engine = "Engine Number";
datalines;
 1 78.4   2 80.1   3 84.4   4 79.1   5 80.4
 6 83.5   7 73.8   8 83.5   9 75.0  10 76.8
11 70.5  12 80.3  13 82.4  14 79.4  15 86.4
16 90.5  17 77.7  18 82.5  19 79.9  20 83.2
;
```

A partial listing of JETS is shown in Figure 34.1.

The Data Set JETS

engine	diam
1	78.4
2	80.1
3	84.4
4	79.1
.	.
.	.
.	.

Figure 34.1. Partial Listing of the Data Set JETS

Each observation contains the diameter measurement and identification number for a particular engine. The variable ENGINE identifies the sequence of engines and is referred to as the *subgroup-variable*.* The variable DIAM contains the measurements and is referred to as the *process variable* (or *process* for short).

Since the production rate is low, individual measurements and moving range charts are used to monitor the process. The following statements create the charts shown in Figure 34.2:

*Technically, the data for individual measurements and moving range charts are not arranged in rational subgroups. The term *subgroup-variable* is used for consistency with other chart statements in the SHEWHART procedure, and it is convenient to think of the "subgroups" as consisting of single measurements.

```
title 'Individual Measurements and Moving Range Charts';
title2 'Jet Engine Diameters (cm)';
symbol v=dot;
proc shewhart data=jets;
   irchart diam*engine;
run;
```

This example illustrates the basic form of the IRCHART statement. After the keyword IRCHART, you specify the *process* to analyze (in this case, DIAM), followed by an asterisk and the *subgroup-variable* (ENGINE).

The input data set is specified with the DATA= option in the PROC SHEWHART statement.



Figure 34.2. Individual Measurements and Moving Range Charts

Each point on the individual measurements chart indicates the inner diameter of a particular engine. Each point on the moving range chart indicates the range of the two most recent measurements. For instance, the moving range plotted for the second engine is $|78.4 - 80.1| = 1.7$. No moving range is plotted for the first engine. Since all of the individual measurements and moving ranges lie within the control limits, it can be concluded that the process is in statistical control.

By default, the control limits shown are 3σ limits estimated from the data; the formulas for the limits are given at "34.22" on page 1165. You can also read control limits from an input data set; see "Reading Preestablished Control Limits" on page 1150.

Saving Individual Measurements and Moving Ranges

See SHWIR1 in the SAS/QC Sample Library

In this example, the IRCHART statement is used to create an output data set containing individual measurements and moving ranges. The following statements read the diameter measurements from the data set JETS (see page 1144) and create a data set named JETINFO:

```
title 'Individual Measurements and Moving Ranges for Diameters';
proc shewhart data=jets;
   irchart diam*engine / outhistory = jetinfo
                         nochart;
run;
```

The OUTHISTORY= option names the output data set, and the NOCHART option suppresses the display of the charts, which would be identical to those in Figure 34.2. Options such as OUTHISTORY= and NOCHART are specified after the slash (/) in the IRCHART statement. A complete list of options is presented in the "Syntax" section on page 1153.

Figure 34.3 contains a partial listing of JETINFO.

Individual Measurements and Moving Ranges for Diameters

engine	diam	diamR
1	78.4	.
2	80.1	1.7
3	84.4	4.3
4	79.1	5.3
5	80.4	1.3
.	.	.
.	.	.
.	.	.

Figure 34.3. The Data Set JETINFO

The data set JETINFO contains one observation for each engine, and it includes three variables.

- ENGINE contains the subgroup index.
- DIAM contains the individual measurements.
- DIAMR contains the moving ranges.

Note that the variable containing the moving ranges is named by adding the suffix character *R* to the *process* DIAM specified in the IRCHART statement.

For more information, see "OUTHISTORY= Data Set" on page 1167.

Reading Individual Measurements and Moving Ranges

See SHWIR1 in the SAS/QC Sample Library

In some applications, both individual measurements and moving ranges may be provided. You can read this type of data set by specifying it with the HISTORY= option in the PROC SHEWHART statement. For example, the following statements read the data set JETINFO (see page 1146) and create the charts shown in Figure 34.4:

```
title 'Individual Measurements and Moving Range Charts';
proc shewhart history=jetinfo lineprinter;
   irchart diam*engine='*';
run;
```

Note that the charts are produced on a line printer since the LINEPRINTER option is specified in the PROC SHEWHART statement. * The asterisk (*) specified in single quotes after the *subgroup-variable* indicates the character used to plot points. This character must follow an equal sign.



Figure 34.4. Charts from the Summary Data Set JETINFO

*In Release 6.12 and previous releases of SAS/QC software, the keyword GRAPHICS was required in the PROC SHEWHART statement to specify that the chart be created with a graphics device. In Version 7, you can specify the LINEPRINTER option to request line printer plots.

A HISTORY= data set used with the IRCHART statement must contain the following variables:

- subgroup variable
- individual measurements variable
- moving range variable

Furthermore, the name of the moving range variable must begin with the *process* name specified in the IRCHART statement and end with the special suffix character *R*. If the name does not follow this convention, you can use the RENAME option in the PROC SHEWHART statement to rename this variable for the duration of the procedure step (see page 1507). For more information, see "HISTORY= Data Set" on page 1170.

Saving Control Limits

See SHWIR1 in the SAS/QC Sample Library

You can save the control limits for individual measurements and moving range charts in a SAS data set; this enables you to apply the control limits to future data (see "Reading Preestablished Control Limits" on page 1150) or modify the limits with a DATA step program.

The following statements read the diameter measurements from the data set JETS (see page 1144) and save the control limits displayed in Figure 34.2 in a data set named JETLIM:

```
title 'Control Limits for Diameters';
proc shewhart data=jets;
   irchart diam*engine / outlimits = jetlim
                         nochart;
run;
```

The OUTLIMITS= option names the data set containing the control limits, and the NOCHART option suppresses the display of the charts. The data set JETLIM is listed in Figure 34.5.

Control Limits for Diameters

VAR	_SUBGRP_	_TYPE_	_LIMITN_	_ALPHA_	_SIGMAS_	_LCLI_
diam	engine	ESTIMATE	2	.002699796	3	66.2290

MEAN	_UCLI_	_LCLR_	_R_	_UCLR_	_STDDEV_
80.39	94.5510	0	5.32632	17.3986	4.72032

Figure 34.5. The Data Set JETLIM Containing Control Limit Information

The data set JETLIM contains one observation with the limits for *process* DIAM. The variables _LCLI_ and _UCLI_ contain the control limits for the individual measurements, and the variable _MEAN_ contains the central line. The variables _LCLR_ and _UCLR_ contain the control limits for the moving ranges, and the variable _R_ contains the central line. The value of _MEAN_ is an estimate of the process mean, and the value of _STDDEV_ is an estimate of the process standard deviation σ. The

value of _LIMITN_ is the number of consecutive measurements used to compute the moving ranges, and the value of _SIGMAS_ is the multiple of σ associated with the control limits. The variables _VAR_ and _SUBGRP_ are bookkeeping variables that save the *process* and *subgroup-variable*. The variable _TYPE_ is a bookkeeping variable that indicates whether the values of _MEAN_ and _STDDEV_ are estimates or standard values. For more information, see "OUTLIMITS= Data Set" on page 1166.

You can create an output data set containing both control limits and summary statistics with the OUTTABLE= option, as illustrated by the following statements:

```
title 'Summary Statistics and Control Limit Information';
proc shewhart data=jets;
   irchart diam*engine / outtable=jtable
                         nochart;
run;
```

The data set JTABLE is listed in Figure 34.6.

Summary Statistics and Control Limit Information

VAR	engine	_SIGMAS_	_LIMITN_	_LCLI_	_SUBI_	_MEAN_	_UCLI_	_EXLIM_	_LCLR_	_SUBR_	_R_	_UCLR_	_EXLIMR_
diam	1	3	2	66.2290	78.4	80.39	94.5510	0	.	5.32632		17.3986	
diam	2	3	2	66.2290	80.1	80.39	94.5510	0	1.7	5.32632		17.3986	
diam	3	3	2	66.2290	84.4	80.39	94.5510	0	4.3	5.32632		17.3986	
diam	4	3	2	66.2290	79.1	80.39	94.5510	0	5.3	5.32632		17.3986	
diam	5	3	2	66.2290	80.4	80.39	94.5510	0	1.3	5.32632		17.3986	
diam	6	3	2	66.2290	83.5	80.39	94.5510	0	3.1	5.32632		17.3986	
diam	7	3	2	66.2290	73.8	80.39	94.5510	0	9.7	5.32632		17.3986	
diam	8	3	2	66.2290	83.5	80.39	94.5510	0	9.7	5.32632		17.3986	
diam	9	3	2	66.2290	75.0	80.39	94.5510	0	8.5	5.32632		17.3986	
diam	10	3	2	66.2290	76.8	80.39	94.5510	0	1.8	5.32632		17.3986	
diam	11	3	2	66.2290	70.5	80.39	94.5510	0	6.3	5.32632		17.3986	
diam	12	3	2	66.2290	80.3	80.39	94.5510	0	9.8	5.32632		17.3986	
diam	13	3	2	66.2290	82.4	80.39	94.5510	0	2.1	5.32632		17.3986	
diam	14	3	2	66.2290	79.4	80.39	94.5510	0	3.0	5.32632		17.3986	
diam	15	3	2	66.2290	86.4	80.39	94.5510	0	7.0	5.32632		17.3986	
diam	16	3	2	66.2290	90.5	80.39	94.5510	0	4.1	5.32632		17.3986	
diam	17	3	2	66.2290	77.7	80.39	94.5510	0	12.8	5.32632		17.3986	
diam	18	3	2	66.2290	82.5	80.39	94.5510	0	4.8	5.32632		17.3986	
diam	19	3	2	66.2290	79.9	80.39	94.5510	0	2.6	5.32632		17.3986	
diam	20	3	2	66.2290	83.2	80.39	94.5510	0	3.3	5.32632		17.3986	

Figure 34.6. The Data Set JTABLE

This data set contains one observation for each subgroup. The variables _SUBI_ and _SUBR_ contain the individual measurements and moving ranges. The variables _LCLI_ and _UCLI_ contain the lower and upper control limits for the individual measurements chart, and the variables _LCLR_ and _UCLR_ contain the lower and upper control limits for the moving range chart. The variable _MEAN_ contains the central line of the individual measurements chart, and the variable _R_ contains

the central line of the moving range chart. The variables _VAR_ and ENGINE contain the *process* name and values of the *subgroup-variable*, respectively. For more information, see “OUTTABLE= Data Set” on page 1168.

An OUTTABLE= data set can be read later as a TABLE= data set. For example, the following statements read JTABLE and display charts (not shown here) identical to those in Figure 34.2:

```
title 'Individual Measurements and Moving Range Control Charts';
title2 'Jet Engine Diameters (cm)';
proc shewhart table=jtable;
   irchart diam*engine;
run;
```

Because the SHEWHART procedure simply displays the information in a TABLE= data set, you can use TABLE= data sets to create specialized control charts (see Chapter 49, “Specialized Control Charts”).

For more information, see “TABLE= Data Set” on page 1171.

Reading Preestablished Control Limits

See SHWIR1 in the SAS/QC Sample Library

In the previous example, the OUTLIMITS= data set JETLIM saved control limits computed from the measurements in JETS. This example shows how these limits can be applied to data for an additional 20 jet engines provided in the following data set:

```
data jets2;
   input engine diam @@;
   label diam   = "Inner Diameter (cm)"
         engine = "Engine Number";
datalines;
21 81.8  22 87.5  23 80.0  24 89.3  25 83.9
26 76.3  27 75.8  28 82.4  29 82.6  30 77.7
31 79.3  32 81.4  33 76.8  34 75.9  35 86.3
36 77.4  37 80.9  38 87.1  39 85.7  40 73.3
;
```

The following statements create individual measurements and moving range charts for the data in JETS2 using the control limits in JETLIM:

```
title 'Individual Measurements and Moving Range Control Charts';
proc shewhart data=jets2 limits=jetlim;
   irchart diam*engine;
run;
```

The charts are shown in Figure 34.7. The LIMITS= option in the PROC SHEWHART statement specifies the data set containing the control limits. By default,* this information is read from the first observation in the LIMITS= data set for which

*In Release 6.09 and in earlier releases, it is also necessary to specify the READLIMITS option to read control limits from a LIMITS= data set.

- the value of _VAR_ matches the *process* name DIAM
- the value of _SUBGRP_ matches the *subgroup-variable* name ENGINE

The charts indicate that the process is in control, since all the individual measurements and moving ranges lie within their respective control limits.

In this example, the LIMITS= data set was created in a previous run of the SHEWHART procedure. You can also create a LIMITS= data set with the DATA step. See "LIMITS= Data Set" on page 1169 for details concerning the variables that you must provide.



Figure 34.7. Charts for Second Set of Engine Noise Levels

Specifying the Computation of the Moving Range

By default, the IRCHART statement uses two consecutive measurements to calculate moving ranges. However, you can specify a different number of measurements to use, as illustrated by the following statements:

See SHWIR1 in the SAS/QC Sample Library

```
title 'Specifying the Computation of the Moving Range';
symbol v=dot;
proc shewhart data=jets;
   irchart diam*engine / limitn=3;
run;
```

The LIMITN= option specifies the number of consecutive measurements used to compute the moving ranges. The resulting charts are shown in Figure 34.8.



Figure 34.8. Computing Moving Ranges from Three Consecutive Measurements

Note that the LIMITN= value is displayed in the legend above the control limit labels. The charts indicate that the process is in control, since all the points lie within the control limits.

Syntax

The basic syntax for the IRCHART statement is as follows:

IRCHART *process******subgroup-variable* **;**

The general form of this syntax is as follows:

IRCHART (*processes***)***subgroup-variable* <**(***block-variables* **)** >
< =*symbol-variable* | =*'character'* > < **/** *options* >**;**

You can use any number of IRCHART statements in the SHEWHART procedure. The components of the IRCHART statement are described as follows.

process
processes
identify one or more processes to be analyzed. The specification of *process* depends on the input data set specified in the PROC SHEWHART statement.

- If raw data are read from a DATA= data set, *process* must be the name of the variable containing the individual measurements. For an example, see "Creating Individual Measurements and Moving Range Charts" on page 1144.
- If individual measurements and moving ranges are read from a HISTORY= data set, *process* must be the name of the variable containing the individual measurements as well as the prefix of the variable containing the moving ranges in the HISTORY= data set. For an example, see "Saving Individual Measurements and Moving Ranges" on page 1146.
- If individual measurements, moving ranges, and control limits are read from a TABLE= data set, *process* must be the value of the variable _VAR_ in the TABLE= data set. For an example, see "Saving Control Limits" on page 1148.

A *process* is required. If you specify more than one *process*, enclose the list in parentheses. For example, the following statements request distinct individual measurements and moving range charts for WEIGHT, LENGTH, and WIDTH:

```
proc shewhart data=measures;
   irchart (weight length width)*day;
run;
```

subgroup-variable
is the variable that identifies subgroups in the data. The *subgroup-variable* is required. In the preceding IRCHART statement, DAY is the subgroup variable. Note that each "subgroup" consists of a single observation. For details, see "Subgroup Variables" on page 1534.

block-variables
are optional variables that group the data into blocks of consecutive subgroups. The blocks are labeled in a legend, and each *block-variable* provides one level of labels in the legend. See "Displaying Stratification in Blocks of Observations" on page 1684 for an example.

symbol-variable
is an optional variable whose levels (unique values) determine the symbol marker or character used to plot the individual measurements and moving ranges.

- If you produce a chart on a line printer, an 'A' is displayed for the points corresponding to the first level of the *symbol-variable*, a 'B' is displayed for the points corresponding to the second level, and so on.
- If you produce a chart on a graphics device, distinct symbol markers are displayed for points corresponding to the various levels of the *symbol-variable*. You can specify the symbol markers with SYMBOLn statements. See "Displaying Stratification in Levels of a Classification Variable" on page 1683 for an example.

character
specifies a plotting character for charts produced on line printers. For example, the following statements create charts using an asterisk (*) to plot the points:

```
proc shewhart data=values;
   irchart weight*day='*';
run;
```

options
enhance the appearance of the charts, request additional analyses, save results in data sets, and so on. The "Summary of Options" section, which follows, lists all options by function. Chapter 46, "Dictionary of Options," describes each option in detail.

Summary of Options

The following tables list the IRCHART statement options by function. For complete descriptions, see Chapter 46, "Dictionary of Options."

Table 34.1. Tabulation Options

TABLE	creates a basic table of individual measurements, moving ranges, and control limits
TABLEALL	is equivalent to the options TABLE, TABLECENTRAL, TABLEID, TABLELEGEND, TABLEOUT, and TABLETESTS
TABLECENTRAL	augments basic table with values of central lines
TABLEID	augments basic table with columns for ID variables
TABLELEGEND	augments basic table with legend for tests for special causes
TABLEOUTLIM	augments basic table with columns indicating control limits exceeded
TABLETESTS	augments basic table with a column indicating which tests for special causes are positive

Note that specifying (EXCEPTIONS) after a tabulation option creates a table for exceptional points only.

Table 34.2. Options for Specifying Tests for Special Causes

NO3SIGMACHECK	allows tests to be applied with control limits other than 3σ limits
TESTS=*value-list*|*customized-pattern-list*	specifies tests for special causes for the individual measurements chart
TEST2RUN=*n*	specifies length of pattern for Test 2
TEST3RUN=*n*	specifies length of pattern for Test 3
TESTACROSS	applies tests across *phase* boundaries
TESTLABEL='*label*'|(*variable*)|*keyword*	provides labels for points where test is positive
TESTLABEL*n*='*label*'	specifies label for n^{th} test for special causes
TESTNMETHOD=STANDARDIZE	applies tests to standardized chart statistics
TESTOVERLAP	performs tests on overlapping patterns of points
ZONELABELS	adds labels A, B, and C to zone lines on individual measurements chart
ZONES	adds lines to individual measurements chart delineating zones A, B, and C
ZONEVALPOS=*n*	specifies position of ZONEVALUES labels
ZONEVALUES	labels zone lines with their values

Table 34.3. Graphical Options for Displaying Tests for Special Causes

CTESTS=*color*|*test-color-list*	specifies color for labels used to identify points where test is positive
CZONES=*color*	specifies color for lines and labels delineating zones A, B, and C
LABELFONT=*font*	specifies software font for labels at points where test is positive (alias for the TESTFONT= option)
LABELHEIGHT=*value*	specifies height of labels at points where test is positive (alias for the TESTHEIGHT= option)
LTESTS=*linetype*	specifies type of line connecting points where test is positive
LZONES=*linetype*	specifies line type for lines delineating zones A, B, and C
TESTFONT=*font*	specifies software font for labels at points where test is positive
TESTHEIGHT=*value*	specifies height of labels at points where test is positive

Table 34.4. Line Printer Options for Displaying Tests for Special Causes

TESTCHAR='*character*'	specifies character for line segments that connect any sequence of points for which a test for special causes is positive
ZONECHAR='*character*'	specifies character for lines that delineate zones for tests for special causes

Table 34.5. Clipping Options

CCLIP=*color*	specifies color for plot symbol for clipped points
CLIPCHAR='*character*'	specifies plot character for clipped points
CLIPFACTOR=*value*	determines extent to which extreme points are clipped
CLIPLEGEND='*string*'	specifies text for clipping legend
CLIPLEGPOS=*keyword*	specifies position of clipping legend
CLIPSUBCHAR='*character*'	specifies substitution character for CLIPLEGEND= text
CLIPSYMBOL=*symbol*	specifies plot symbol for clipped points
CLIPSYMBOLHT=*value*	specifies symbol marker height for clipped points

Table 34.6. Options for Plotting and Labeling Points

ALLLABEL=VALUE|(*variable*)	labels every point on individual measurements chart
ALLLABEL2=VALUE|(*variable*)	labels every point on moving range chart
CCONNECT=*color*	specifies color for line segments that connect points on chart
CFRAMELAB=*color*	specifies fill color for frame around labeled points
CNEEDLES=*color*	specifies color for needles that connect points to central line
CONNECTCHAR='*character*'	specifies character used to form line segments that connect points on chart
COUT=*color*	specifies color for portions of line segments that connect points outside control limits
COUTFILL=*color*	specifies color for shading areas between the connected points and control limits outside the limits
NEEDLES	connects points to central line with vertical needles
NOCONNECT	suppresses line segments that connect points on chart
OUTLABEL=VALUE|(*variable*)	labels points outside control limits on individual measurements chart
OUTLABEL2=VALUE|(*variable*)	labels points outside control limits on moving range chart
SYMBOLCHARS='*characters*'	specifies characters indicating *symbol-variable*
SYMBOLLEGEND=NONE|*name*	specifies LEGEND statement for levels of *symbol-variable*
SYMBOLORDER=*keyword*	specifies order in which symbols are assigned for levels of *symbol-variable*
TURNALL|TURNOUT	turns point labels so that they are strung out vertically

Table 34.7. Reference Line Options

CHREF=*color*	specifies color for lines requested by the HREF= and HREF2= options
CVREF=*color*	specifies color for lines requested by the VREF= and VREF2= options
HREF=*values*|*SAS-data-set*	specifies position of reference lines perpendicular to horizontal axis on individual measurements chart
HREF2=*values*|*SAS-data-set*	specifies position of reference lines perpendicular to horizontal axis on moving range chart
HREFCHAR='*character*'	specifies line character for HREF= and HREF2= lines
HREFDATA=*SAS-data-set*	specifies position of reference lines perpendicular to horizontal axis on individual measurements chart
HREF2DATA=*SAS-data-set*	specifies position of reference lines perpendicular to horizontal axis on moving range chart
HREFLABELS=('*label1*'...'*labeln*')	specifies labels for HREF= lines
HREF2LABELS=('*label1*'...'*labeln*')	specifies labels for HREF2= lines
HREFLABPOS=*n*	specifies position of HREFLABELS= and HREF2LABELS= labels
LHREF=*linetype*	specifies line type for HREF= and HREF2= lines
LVREF=*linetype*	specifies line type for VREF= and VREF2= lines
NOBYREF	specifies that reference line information in a data set is to be applied uniformly to charts created for all BY groups
VREF=*values*|*SAS-data-set*	specifies position of reference lines perpendicular to vertical axis on individual measurements chart
VREF2=*values*|*SAS-data-set*	specifies position of reference lines perpendicular to vertical axis on moving range chart
VREFCHAR='*character*'	specifies line character for VREF= and VREF2= lines
VREFLABELS='*label1*'...'*labeln*'	specifies labels for VREF= lines
VREF2LABELS='*label1*'...'*labeln*'	specifies labels for VREF2= lines
VREFLABPOS=*n*	specifies position of VREFLABELS= and VREF2LABELS= labels

Table 34.8. Block Variable Legend Options

BLOCKLABELPOS=*keyword*	specifies position of label for *block-variable* legend
BLOCKLABTYPE=*n*|*keyword*	specifies text size of *block-variable* legend
BLOCKPOS=*n*	specifies vertical position of *block-variable* legend
BLOCKREP	repeats identical consecutive labels in *block-variable* legend
CBLOCKLAB=*color*	specifies color for filling background in *block-variable* legend
CBLOCKVAR=*variable*|(*variables*)	specifies one or more variables whose values are colors for filling background of *block-variable* legend

Table 34.9. Process Mean and Standard Deviation Options

MU0=*value*	specifies known value μ_0 for process mean μ
SIGMA0=*value*	specifies known value σ_0 for process standard deviation σ
SMETHOD=*keyword*	specifies method for estimating process standard deviation σ
TYPE=*keyword*	identifies whether parameters are estimates or standard values and specifies value of _TYPE_ in OUTLIMITS= data set

Table 34.10. Phase Options

CPHASEBOX=*color*	specifies color for box enclosing all plotted points for a phase
CPHASEBOX-CONNECT=*color*	specifies color for line segments connecting adjacent enclosing boxes
CPHASEBOXFILL=*color*	specifies fill color for box enclosing all plotted points for a phase
CPHASELEG=*color*	specifies text color for *phase* legend
CPHASEMEAN-CONNECT=*color*	specifies color for line segments connecting average value points within a phase
NOPHASEFRAME	suppresses default frame for *phase* legend
OUTPHASE='*string*'	specifies value of _PHASE_ in the OUTHISTORY= data set
PHASEBREAK	disconnects last point in a *phase* from first point in next *phase*
PHASELABTYPE=*value*|*keyword*	specifies text size of *phase* legend
PHASELEGEND	displays *phase* labels in a legend across top of chart
PHASELIMITS	labels control limits for each phase, provided they are constant within that phase
PHASEMEANSYMBOL=*symbol*	specifies symbol marker for average of values within a phase
PHASEREF	delineates *phases* with vertical reference lines
READPHASES= ALL|'*label1*'...'*labeln*'	specifies *phases* to be read from an input data set

Table 34.11. Axis and Axis Label Options

CAXIS=*color*	specifies color for axis lines and tick marks
CFRAME=*color*|(*color-list*)	specifies fill colors for frame for plot area
CTEXT=*color*	specifies color for tick mark values and axis labels
HAXIS=*values*|AXISn	specifies major tick mark values for horizontal axis
HEIGHT=*value*	specifies height of axis label and axis legend text
HMINOR=*n*	specifies number of minor tick marks between major tick marks on horizontal axis
HOFFSET=*value*	specifies length of offset at both ends of horizontal axis
INTSTART=*value*	specifies first major tick mark value for numeric horizontal axis
NOHLABEL	suppresses label for horizontal axis
NOTICKREP	specifies that only the first occurrence of repeated, adjacent subgroup values is to be labeled on horizontal axis
NOTRUNC	suppresses vertical axis truncation at zero applied by default to moving range chart
NOVANGLE	requests vertical axis labels that are strung out vertically
SKIPHLABELS=*n*	specifies thinning factor for tick mark labels on horizontal axis
SPLIT='*character*'	specifies splitting character for axis labels
TURNHLABELS	requests horizontal axis labels that are strung out vertically
VAXIS=*values*|AXISn	specifies major tick mark values for vertical axis of individual measurements chart
VAXIS2=*values*|AXISn	specifies major tick mark values for vertical axis of moving range chart
VMINOR=*n*	specifies number of minor tick marks between major tick marks on vertical axis
VOFFSET=*value*	specifies length of offset at both ends of vertical axis
VZERO	forces origin to be included in vertical axis for primary chart
VZERO2	forces origin to be included in vertical axis for secondary chart
WAXIS=*n*	specifies width of axis lines

Table 34.12. Options for Interactive Control Charts

HTML=(*variable*)	specifies a variable whose values are URLs to be associated with subgroups
HTML_LEGEND=(*variable*)	specifies a variable whose values are URLs to be associated with symbols in the symbol legend
TESTURLS=*SAS-data-set*	associates URLs with tests for special causes
WEBOUT=*SAS-data-set*	creates an OUTTABLE= data set with additional graphics coordinate data

Table 34.13. Options for Displaying Control Limits

CINFILL=*color*	specifies color for area inside control limits
CLIMITS=*color*	specifies color of control limits, central line, and related labels
LCLLABEL='*label*'	specifies label for lower control limit on individual measurements chart
LCLLABEL2='*label*'	specifies label for lower control limit on moving range chart
LIMLABSUBCHAR='*character*'	specifies a substitution character for labels provided as quoted strings; the character is replaced with the value of the control limit
LLIMITS=*linetype*	specifies line type for control limits
NDECIMAL=*n*	specifies number of digits to right of decimal place in default labels for control limits and central line on individual measurements chart
NDECIMAL2=*n*	specifies number of digits to right of decimal place in default labels for control limits and central line on moving range chart
NOCTL	suppresses display of central line on individual measurements chart
NOCTL2	suppresses display of central line on moving range chart
NOLCL	suppresses display of lower control limit on individual measurements chart
NOLCL2	suppresses display of lower control limit on moving range chart
NOLIMITLABEL	suppresses labels for control limits and central lines
NOLIMITS	suppresses display of control limits
NOLIMITSFRAME	suppresses default frame around control limit information when multiple sets of control limits are read from LIMITS= data set
NOLIMITSLEGEND	suppresses legend for control limits
NOLIMIT0	suppresses display of zero lower control limit on moving range chart
NOUCL	suppresses display of upper control limit on individual measurements chart
NOUCL2	suppresses display of upper control limit on moving range chart
RSYMBOL='*string*'|*keyword*	specifies label for central line on R chart
UCLLABEL='*string*'	specifies label for upper control limit on individual measurements chart
UCLLABEL2='*string*'	specifies label for upper control limit on moving range chart
WLIMITS=*n*	specifies width for control limits and central line
XSYMBOL='*string*'|*keyword*	specifies label for central line on individual measurements chart

Table 34.14. Options for Specifying Control Limits

ALPHA=*value*	requests probability limits for control charts
LIMITN=*n*	specifies number of consecutive measurements used to compute moving ranges
NOREADLIMITS	computes control limits for each *process* from the data rather than from a LIMITS= data set (Release 6.10 and later releases)
READALPHA	reads _ALPHA_ instead of _SIGMAS_ from a LIMITS= data set
READINDEXES=ALL | '*label1*'...'*labeln*'	reads multiple sets of control limits for each *process* from a LIMITS= data set
READLIMITS	reads single set of control limits for each *process* from a LIMITS= data set (Release 6.09 and earlier releases)
SIGMAS=*k*	specifies width of control limits in terms of multiple k of standard error of plotted statistic

Table 34.15. Plot Layout Options

BILEVEL	creates control charts using half-screens and half-pages
EXCHART	creates control charts only when exceptions occur
INTERVAL=*keyword*	specifies natural time interval between consecutive subgroup positions when time, date, or datetime format is used
MAXPANELS=*n*	specifies maximum number of pages or screens for chart
NOCHART	suppresses creation of charts
NOCHART2	suppresses creation of moving range chart
NOFRAME	suppresses frame for plot area
NPANELPOS=*n*	specifies number of subgroup positions per panel on each chart
REPEAT	repeats last subgroup position on panel as first subgroup position of next panel
SEPARATE	displays individual measurements and moving range charts on separate screens or pages
TOTPANELS=*n*	specifies number of pages or screens to be used to display chart
YPCT1=*value*	specifies length of vertical axis on individual measurements chart as a percentage of sum of lengths of vertical axes for individual measurements and moving range charts
ZEROSTD	displays individual measurements and moving range charts regardless of whether $\hat{\sigma} = 0$

Table 34.16. Graphical Enhancement Options

ANNOTATE= *SAS-data-set*	specifies annotate data set that adds features to individual measurements chart
ANNOTATE2= *SAS-data-set*	specifies annotate data set that adds features to moving range chart
DESCRIPTION='*string*'	specifies string that appears in the description field of the PROC GREPLAY master menu for individual measurements chart
DESCRIPTION2='*string*'	specifies string that appears in the description field of the PROC GREPLAY master menu for moving range chart
FONT=*font*	specifies software font for labels and legends on charts
LTMARGIN=*value*	specifies width of left margin area for plot requested with LTMPLOT= option
LTMPLOT=*keyword*	requests univariate plot in left margin
NAME='*string*'	specifies name that appears in the name field of the PROC GREPLAY master menu for individual measurements chart
NAME2='*string*'	specifies name that appears in the name field of the PROC GREPLAY master menu for moving range chart
PAGENUM='*string*'	specifies the form of the label used in pagination
PAGENUMPOS= *keyword*	specifies the position of the page number requested with the PAGENUM= option
RTMARGIN=*value*	specifies width of right margin area for plot requested with RTMPLOT= option
RTMPLOT=*keyword*	requests univariate plot in right margin

Table 34.17. Grid Options

ENDGRID	adds grid after last plotted point
GRID	adds grid to control chart
LENDGRID=*linetype*	specifies line type for grid requested with the ENDGRID option
LGRID=*linetype*	specifies line type for grid requested with the GRID option
WGRID=*n*	specifies width of grid lines

Table 34.18. Specification Limit Options

CIINDICES=(ALPHA=*value* TYPE=*keyword*)	specifies α value and type for computing capability index confidence limits
LSL=*value-list*	specifies list of lower specification limits
TARGET=*value-list*	specifies list of target values
USL=*value-list*	specifies list of upper specification limits

Table 34.19. Input Data Set Options

MISSBREAK	specifies that observations with missing values are not to be processed

Table 34.20. Output Data Set Options

OUTHISTORY= *SAS-data-set*	creates output data set containing individual measurements and moving ranges
OUTINDEX='*string*'	specifies value of _INDEX_ in the OUTLIMITS= data set
OUTLIMITS= *SAS-data-set*	creates output data set containing control limits
OUTTABLE= *SAS-data-set*	creates output data set containing individual measurements, moving ranges, and control limits

Table 34.21. Star Options

CSTARCIRCLES=*color*	specifies color for STARCIRCLES= circles
CSTARFILL=*color*| *(variable)*	specifies color for filling stars
CSTAROUT=*color*	specifies outline color for stars exceeding inner or outer circles
CSTARS=*color*| *(variable)*	specifies color for outlines of stars
LSTARCIRCLES= *linetypes*	specifies line types for STARCIRCLES= circles
LSTARS=*linetype*|*(variable)*	specifies line types for outlines of STARVERTICES= stars
STARBDRADIUS=*value*	specifies radius of outer bound circle for vertices of stars
STARCIRCLES=*value-list*	specifies reference circles for stars
STARINRADIUS=*value*	specifies inner radius of stars
STARLABEL=*keyword*	specifies vertices to be labeled
STARLEGEND=*keyword*	specifies style of legend for star vertices
STARLEGENDLAB='*label*'	specifies label for STARLEGEND= legend
STAROUTRADIUS=*value*	specifies outer radius of stars
STARSPEC=*value*| *SAS-data-set*	specifies method used to standardize vertex variables
STARSTART=*value*	specifies angle for first vertex
STARTYPE=*keyword*	specifies graphical style of star
STARVERTICES= *variable*|*(variables)*	superimposes star at each point on individual measurements chart
WSTARCIRCLES=*n*	specifies width of STARCIRCLES= circles
WSTARS=*n*	specifies width of STARVERTICES= stars

Details

Constructing Charts for Individual Measurements and Moving Ranges

The following notation is used in this section:

μ	process mean (expected value of the population of measurements)
σ	process standard deviation (standard deviation of the population of measurements)
X_i	the i^{th} individual measurement
$\bar{X}$	mean of the individual measurements, computed as $(X_1 + \cdots + X_N)/N$, where N is the number of individual measurements
n	number of consecutive measurements used to calculate the moving ranges (by default, $n = 2$)
R_i	moving range computed for the i^{th} subgroup (corresponding to the i^{th} individual measurement). If $i < n$, then R_i is assigned a missing value. Otherwise, $$R_i = \max(X_i, X_{i-1}, ..., X_{i-n+1}) - \min(X_i, X_{i-1}, ..., X_{i-n+1})$$ This formula assumes that $X_i, X_{i-1}, ..., X_{i-n+1}$ are nonmissing.
$\bar{R}$	average of the nonmissing moving ranges, computed as $$\frac{R_n + R_{n+1} \cdots + R_N}{N + 1 - n}$$
$d_2(n)$	expected value of the range of n independent normally distributed variables with unit standard deviation
$d_3(n)$	standard error of the range of n independent observations from a normal population with unit standard deviation
z_p	$100p^{\text{th}}$ percentile $(0 < p < 1)$ of the standard normal distribution
$D_p(n)$	$100p^{\text{th}}$ percentile $(0 < p < 1)$ of the distribution of the range of n independent observations from a normal population with unit standard deviation

Plotted Points

Each point on an individual measurements chart, indicates the value of a measurement (X_i).

Each point on a moving range chart indicates the value of a moving range (R_i). With $n = 2$, for example, if the first three measurements are 3.4, 3.7, and 3.6, the first moving range is missing, the second moving range is $|3.7 - 3.4| = 0.3$, and the third moving range is $|3.6 - 3.7| = 0.1$.

Central Lines

By default, the central line on an individual measurements chart indicates an estimate for μ, which is computed as $\bar{X}$. If you specify a known value (μ_0) for μ, the central line indicates the value of μ_0.

The central line on a moving range chart indicates an estimate for the expected moving range, computed as $d_2(n)\hat{\sigma}$ where $\hat{\sigma} = \bar{R}/d_2(n)$. If you specify a known value ($\hat{\sigma}_0$) for σ, the central line indicates the value of $d_2(n)\sigma_0$.

Control Limits

You can compute the limits

- as a specified multiple (k) of the standard errors of X_i and R_i above and below the central line. The default limits are computed with $k = 3$ (these are referred to as 3σ *limits*).
- as probability limits defined in terms of α, a specified probability that X_i or R_i exceeds the limits

The following table provides the formulas for the limits:

Table 34.22. Limits for Individual Measurements and Moving Range Charts

Control Limits	
Individual Measurements Chart	LCL = lower control limit = $\bar{\bar{X}} - k\hat{\sigma}$ UCL = upper control limit = $\bar{\bar{X}} + k\hat{\sigma}$
Moving Range Chart	LCL = lower control limit = $\max(d_2(n)\hat{\sigma} - kd_3(n)\hat{\sigma}, 0)$ UCL = upper control limit = $d_2(n)\hat{\sigma} + kd_3(n)\hat{\sigma}$

Probability Limits	
Individual Measurements Chart	LCL = lower control limit = $\bar{\bar{X}} - z_{\alpha/2}\hat{\sigma}$ UCL = upper control limit = $\bar{\bar{X}} + z_{\alpha/2}\hat{\sigma}$
Moving Range Chart	LCL = lower control limit = $D_{\alpha/2}(n)\hat{\sigma}$ UCL = upper control limit = $D_{1-\alpha/2}(n)\hat{\sigma}$

The formulas assume that the measurements are normally distributed. Note that the probability limits for the moving range are asymmetric about the central line. If standard values μ_0 and σ_0 are available for μ and σ, replace $\bar{\bar{X}}$ with μ_0 and $\hat{\sigma}$ with σ_0 in Table 34.22.

You can specify parameters for the limits as follows:

- Specify k with the SIGMAS= option or with the variable _SIGMAS_ in a LIMITS= data set.
- Specify α with the ALPHA= option or with the variable _ALPHA_ in a LIMITS= data set.
- Specify n with the LIMITN= option or with the variable _LIMITN_ in a LIMITS= data set.
- Specify μ_0 with the MU0= option or with the variable _MEAN_ in the LIMITS= data set.
- Specify σ_0 with the SIGMA0= option or with the variable _STDDEV_ in the LIMITS= data set.

Output Data Sets

OUTLIMITS= Data Set

The OUTLIMITS= data set saves control limits and control limit parameters. The following variables can be saved:

Table 34.23. OUTLIMITS= Data Set

Variable	Description
ALPHA	probability (α) of exceeding limits
_CP_	capability index C_p
CPK	capability index C_{pk}
CPL	capability index CPL
CPM	capability index C_{pm}
CPU	capability index CPU
INDEX	optional identifier for the control limits specified with the OUTINDEX= option
LCLI	lower control limit for individual measurements
LCLR	lower control limit for moving ranges
LIMITN	number of consecutive measurements used to compute moving ranges
LSL	lower specification limit
MEAN	process mean
_R_	value of central line on moving range chart
SIGMAS	multiple (k) of standard error of individual measurement or moving range
STDDEV	process standard deviation ($\hat{\sigma}$ or σ_0)
SUBGRP	*subgroup-variable* specified in the IRCHART statement
TARGET	target value
TYPE	type (estimate or standard value) of _MEAN_ and _STDDEV_
UCLI	upper control limit for individual measurements
UCLR	upper control limit for moving ranges range
USL	upper specification limit
VAR	*process* specified in the IRCHART statement

Notes:

1. If the limits are defined in terms of a multiple k of the standard errors of X_i and R_i, the value of _ALPHA_ is computed as $\alpha = 2(1 - \Phi(k))$, where $\Phi(\cdot)$ is the standard normal distribution function.

2. If the limits are probability limits, the value of _SIGMAS_ is computed as $k = \Phi^{-1}(1 - \alpha/2)$, where Φ^{-1} is the inverse standard normal distribution function.

3. The variables _CP_, _CPK_, _CPL_, _CPU_, _LSL_, and _USL_ are included only if you provide specification limits with the LSL= and USL= options. The variables _CPM_ and _TARGET_ are included if, in addition, you provide a target value with the TARGET= option. See "Capability Indices" on page 1537 for computational details.

4. Optional BY variables are saved in the OUTLIMITS= data set.

The OUTLIMITS= data set contains one observation for each *process* specified in the IRCHART statement. For an example, see "Saving Control Limits" on page 1148.

OUTHISTORY= Data Set

The OUTHISTORY= data set saves individual measurements and moving ranges. The following variables are saved:

- the *subgroup-variable*
- an individual measurements variable named by *process*
- a moving range variable named by *process* suffixed with *R*

Given a *process* name that contains eight characters, the procedure first shortens the name to its first four characters and its last three characters, and then it adds the suffix. For example, the procedure shortens the *process* DIAMETER to DIAMTER before adding the suffix *R*.

A variable containing the moving ranges is created for each *process* specified in the IRCHART statement. For example, consider the following statements:

```
proc shewhart data=steel;
   irchart (width diameter)*lot / outhistory=summary;
run;
```

The data set SUMMARY contains variables named LOT, WIDTH, WIDTHR, DIAMETER, and DIAMTERR.

Additionally, the following variables, if specified, are included:

- BY variables
- *block-variables*
- *symbol-variable*
- ID variables
- _PHASE_ (if the OUTPHASE= option is specified)

For an example of an OUTHISTORY= data set, see "Saving Individual Measurements and Moving Ranges" on page 1146.

OUTTABLE= Data Set

The OUTTABLE= data set saves individual measurements, moving ranges, control limits, and related information. The following variables are saved:

Variable	Description
ALPHA	probability (α) of exceeding control limits
EXLIM	control limit exceeded on individual measurements chart
EXLIMR	control limit exceeded on moving range chart
LCLI	lower control limit for individual measurements
LCLR	lower control limit for moving range
LIMITN	number of consecutive measurements used to compute moving ranges
MEAN	process mean
_R_	average range
SIGMAS	multiple (k) of the standard error associated with control limits
subgroup	values of the subgroup variable
SUBI	individual measurement
SUBR	moving range
TESTS	tests for special causes signaled on individual measurements chart
UCLI	upper control limit for individual measurements
UCLR	upper control limit for moving range
VAR	*process* specified in the IRCHART statement

In addition, the following variables, if specified, are included:

- BY variables
- *block-variables*
- *symbol-variable*
- ID variables
- _PHASE_ (if the READPHASES= option is specified)

Notes:

1. Either the variable _ALPHA_ or the variable _SIGMAS_ is saved, depending on how the control limits are defined (with the ALPHA= or SIGMAS= options, respectively, or with the corresponding variables in a LIMITS= data set).

2. The variable _TESTS_ is saved if you specify the TESTS= option. The k^{th} character of a value of _TESTS_ is k if Test k is positive at that subgroup. For example, if you request all eight tests and Tests 2 and 8 are positive for a given subgroup, the value of _TESTS_ has a 2 for the second character, an 8 for the eighth character, and blanks for the other six characters.

3. The variables _VAR_, _EXLIM_, _EXLIMR_, and _TESTS_ are character variables of length 8. The variable _PHASE_ is a character variable of length 16. All other variables are numeric.

For an example, see “Saving Control Limits” on page 1148.

ODS Tables

The following table summarizes the ODS tables that you can request with the IRCHART statement.

Table 34.24. ODS Tables Produced with the IRCHART Statement

Table Name	Description	Options
IRCHART	individual measurement and moving range chart summary statistics	TABLE, TABLEALL, TABLEC, TABLEID, TABLELEG, TABLEOUT, TABLETESTS
Tests	descriptions of tests for special causes requested with the TESTS= option for which at least one positive signal is found	TABLEALL, TABLELEG

Input Data Sets

DATA= Data Set

You can read individual measurements from a DATA= data set specified in the PROC SHEWHART statement. Each *process* specified in the IRCHART statement must be a SAS variable in the data set. This variable provides measurements of items indexed by the *subgroup-variable*. The *subgroup-variable*, which is specified in the IRCHART statement, must also be a SAS variable in the data set. Each observation in a DATA= data set must contain a measurement for each *process* and a value for the *subgroup-variable*. Other variables that can be read from a DATA= data set include

- _PHASE_ (if the READPHASES= option is specified)
- *block-variables*
- *symbol-variable*
- BY variables
- ID variables

By default, the SHEWHART procedure reads all of the observations in a DATA= data set. However, if the DATA= data set includes the variable _PHASE_, you can read selected groups of observations (referred to as *phases*) by specifying the READPHASES= option in the IRCHART statement (for an example, see "Displaying Stratification in Phases" on page 1689).

For an example of a DATA= data set, see "Creating Individual Measurements and Moving Range Charts" on page 1144.

LIMITS= Data Set

You can read preestablished control limits (or parameters from which the control limits can be calculated) from a LIMITS= data set specified in the PROC SHEWHART statement. For example, the following statements read control limit information from the data set CONLIMS:*

*In Release 6.09 and in earlier releases, it is necessary to specify the READLIMITS option.

```
proc shewhart data=info limits=conlims;
   irchart weight*id;
run;
```

The LIMITS= data set can be an OUTLIMITS= data set that was created in a previous run of the SHEWHART procedure. Such data sets always contain the variables required for a LIMITS= data set; see Table 34.22 on page 1165. The LIMITS= data set can also be created directly using a DATA step.

When you create a LIMITS= data set, you must provide one of the following:

- the variables _LCLI_, _MEAN_, _UCLI_, _LCLR_, _R_, and _UCLR_, which specify the control limits directly
- the variables _MEAN_ and _STDDEV_, which are used to calculate the control limits according to the equations in Table 34.22 on page 1165

In addition, note the following:

- The variables _VAR_ and _SUBGRP_ are required. These must be character variables of length 8.
- The variable _INDEX_ is required if you specify the READINDEX= option; this must be a character variable of length 16.
- The variables _LIMITN_, _SIGMAS_ (or _ALPHA_), and _TYPE_ are optional, but they are recommended to maintain a complete set of control limit information. The variable _TYPE_ must be a character variable of length 8; valid values are `ESTIMATE`, `STANDARD`, `STDMU`, and `STDSIGMA`. See Example 34.2 on page 1177 for an illustration.
- BY variables are required if specified with a BY statement.

For an example, see "Reading Preestablished Control Limits" on page 1150.

HISTORY= Data Set

You can read individual measurements and moving ranges from a HISTORY= data set specified in the PROC SHEWHART statement. This allows you to reuse OUTHISTORY= data sets that have been created in previous runs of the SHEWHART procedure.

A HISTORY= data set used with the IRCHART statement must contain the following:

- the *subgroup-variable*
- an individual measurements variable for each *process*
- a moving range variable for each *process*

The name of the individual measurements variable must be the *process* specified in the IRCHART statement. The name of the moving range variable must be the prefix *process* concatenated with the special suffix character R. For example, consider the following statements:

```
proc shewhart history=summary;
   irchart (weight yldstren)*id;
run;
```

The data set SUMMARY must include the variables ID, WEIGHT, WEIGHTR, YLDSTREN, and YLDSRENR.

Note that if you specify a *process* name that contains eight characters, the name of the moving range variable must be formed from the first four characters and the last three characters of the *process* name, suffixed with R.

Other variables that can be read from a HISTORY= data set include

- _PHASE_ (if the READPHASES= option is specified)
- *block-variables*
- *symbol-variable*
- BY variables
- ID variables

By default, the SHEWHART procedure reads all of the observations in a HISTORY= data set. However, if the data set includes the variable _PHASE_, you can read selected groups of observations (referred to as *phases*) by specifying the READ-PHASES= option (see "Displaying Stratification in Phases" on page 1689 for an example).

For an example of a HISTORY= data set, see "Reading Individual Measurements and Moving Ranges" on page 1147.

TABLE= Data Set

You can read individual measurements, moving ranges, and control limits from a TABLE= data set specified in the PROC SHEWHART statement. This enables you to reuse an OUTTABLE= data set created in a previous run of the SHEWHART procedure. Because the SHEWHART procedure simply displays the information in a TABLE= data set, you can use TABLE= data sets to create specialized control charts. Examples are provided in Chapter 49, "Specialized Control Charts."

The following table lists the variables required in a TABLE= data set used with the IRCHART statement:

Table 34.25. Variables Required in a TABLE= Data Set

Variable	Description
LCLI	lower control limit for individual measurements
LCLR	lower control limit for moving range
LIMITN	number of consecutive measurements used to calculate moving ranges
MEAN	process mean
_R_	average moving range
subgroup-variable	values of the *subgroup-variable*
SUBI	individual measurements
SUBR	moving ranges
UCLI	upper control limit for individual measurements
UCLR	upper control limit for moving range

Other variables that can be read from a TABLE= data set include

- *block-variables*
- *symbol-variable*
- BY variables
- ID variables
- _PHASE_ (if the READPHASES= option is specified). This variable must be a character variable of length 16.
- _TESTS_ (if the TESTS= option is specified). This variable is used to flag tests for special causes and must be a character variable of length 8.
- _VAR_. This variable is required if more than one *process* is specified or if the data set contains information for more than one *process*. This variable must be a character variable of length 8.

For an example of a TABLE= data set, see "Saving Control Limits" on page 1148.

Methods for Estimating the Standard Deviation

When control limits are computed from the input data, three methods (referred to as default, MAD and MMR) are available for estimating the process standard deviation σ.

Default Method

The default estimate for σ is

$$\hat{\sigma} = \bar{R}/d_2(n)$$

where $\bar{R}$ is the average of the moving ranges, n is the number of consecutive individual measurements used to compute each moving range, and the unbiasing factor $d_2(n)$ is defined so that if the observations are normally distributed, the expected value of R_i is

$$E(R_i) = d_2(n_i)\sigma$$

This method is described in the *ASTM Manual on Presentation of Data and Control Chart Analysis* (1976).

MAD Method

If you specify SMETHOD=MAD, a median absolute deviation estimator is computed for σ, as described by Boyles (1997). It is computed as

$$\hat{\sigma} = \text{median}\{|X_i - \tilde{X}|, 1 \leq i \leq N\}/0.6745$$

where $\tilde{X}$ is the sample median.

MMR Method

If you specify SMETHOD=MMR, a median moving range estimator is computed for σ. This estimator is described by Boyles (1997). It is computed as

$$\hat{\sigma} = \tilde{R}/0.954$$

where $\tilde{R}$ is the median of the nonmissing moving ranges.

Interpreting Charts for Individual Measurements and Moving Ranges

Montgomery (1996) points out that a moving range chart should be interpreted with care because "the moving ranges are correlated, and this correlation may often induce a pattern or runs or cycles on the chart." For this reason Nelson (1982) recommends against plotting the moving ranges. Nelson notes that the assumption of normality is more critical for an individual measurements chart than for an $\bar{X}$ chart. You can use the NOCHART2 option in the IRCHART statement to specify that only the individual measurements chart is to be displayed. See Example 34.3 on page 1179 for an illustration. If, instead, you specify the SEPARATE option, the charts for individual measurements and moving ranges are displayed on separate screens.

An alternative method for creating an individual measurements chart is to use the XCHART statement, which uses an estimate of σ based on moving ranges of two consecutive measurements when the subgroup sample sizes are all equal to one. Note that the XCHART statement displays the control limit legend $n = 1$ to indicate the common subgroup sample size, whereas the IRCHART statement displays a legend that indicates the number of consecutive measurements used to compute the moving ranges (the "pseudo subgroup sample size").

Nelson (1982) explains that the reason for estimating the process standard deviation σ from moving ranges of two consecutive measurements rather than the sample standard deviation of the measurements is that "the moving range of two minimizes inflationary effects on the variability which are caused by trends and oscillations that may be present." Nelson suggests that any moving range that exceeds 3.5 times the average moving range should be removed from the calculation of the average moving range.

Axis Labels

You can specify axis labels by assigning labels to particular variables in the input data set, as summarized in the following table:

Axis	Input Data Set	Variable
Horizontal	all	*subgroup-variable*
Vertical (Individual measurements chart)	DATA=	*process*
Vertical (Individual measurements chart)	HISTORY=	subgroup measurement variable
Vertical (Individual measurements chart)	TABLE=	_SUBI_

You can specify distinct labels for the vertical axes of the individual measurements and moving range charts by breaking the vertical axis into two parts with a split character. Specify the split character with the SPLIT= option. The first part labels the vertical axis of the individual measurements chart, and the second part labels the vertical axis of the moving range chart.

For example, the following sets of statements specify the label *Avg gap in mm* for the vertical axis of the individual measurements chart and the label *Range in mm* for the vertical axis of the moving range chart:

```
proc shewhart data=doors;
   irchart gap*hour / split = '/' ;
   label gap = 'Avg gap in mm/Range in mm';
run;

proc shewhart history=doorhist;
   irchart gap*hour / split = '/' ;
   label gap = 'Avg gap in mm/Range in mm';
run;

proc shewhart table=doortab;
   irchart gap*hour / split = '/' ;
   label _SUBI_ = 'Avg gap in mm/Range in mm';
run;
```

In this example, the label assignments are in effect only for the duration of the procedure step, and they temporarily override any permanent labels associated with the variables.

Missing Values

An observation read from a DATA=, HISTORY=, or TABLE= data set is not analyzed if the value of the subgroup variable is missing. For a particular process variable, an observation read from a DATA= data set is not analyzed if the value of the process variable is missing. Missing values of process variables generally lead to unequal subgroup sample sizes. For a particular process variable, an observation read from a HISTORY= or TABLE= data set is not analyzed if the values of any of the corresponding summary variables are missing.

Examples

This section provides advanced examples of the IRCHART statement.

Example 34.1. Applying Tests for Special Causes

This example illustrates how you can apply tests for special causes to make an individual measurements chart more sensitive to special causes of variation. The following statements create a data set named ENGINES, which contains the weights for 25 jet engines:

See SHWIREX1 in the SAS/QC Sample Library

```
data engines;
   input id weight @@;
   label weight='Engine Weight (lbs)'
         id    ='Engine ID Number';
   datalines;
1711 1270   1712 1258   1713 1248   1714 1260
1715 1263   1716 1260   1717 1259   1718 1240
1719 1260   1720 1246   1721 1238   1722 1253
1723 1249   1724 1245   1725 1251   1726 1252
1727 1249   1728 1274   1729 1258   1730 1268
1731 1248   1732 1295   1733 1243   1734 1253
1735 1258
;
```

Individual measurements and moving range charts are used to monitor the weights. The following statements produce the tables shown in Output 34.1.1 and create the charts shown in Output 34.1.2:

```
title 'Tests for Special Causes Applied to Jet Engine Weights';
symbol v=dot;
proc shewhart data=engines;
   irchart weight*id /
       tests   =1 to 8
       test2run=7
       tabletests
       zonelabels
       ltests  =20;
run;
```

The TESTS= option applies eight tests for special causes, which are described in Chapter 48, "Tests for Special Causes." The TEST2RUN= option specifies the length of the pattern for Test 2. The TABLETESTS option requests a table of individual measurements, moving ranges, and control limits, and it adds a column indicating which measurements tested positive for special causes.

The ZONELABELS option displays zone lines and zone labels on the individual measurements chart. The zones are used to define the tests. The LTESTS= option specifies the line type used to connect the points in a pattern for a test that is signaled.

Output 34.1.1. Tabular Form of Individual Measurements and Moving Range Chart

Tests for Special Causes Applied to Jet Engine Weights

Individual Measurements Chart Summary for weight

id	--3 Sigma Limits with n=2 for weight-- Lower Limit	weight	Upper Limit	Special Tests Signaled
1711	1220.4709	1270.0000	1291.3691	
1712	1220.4709	1258.0000	1291.3691	
1713	1220.4709	1248.0000	1291.3691	
1714	1220.4709	1260.0000	1291.3691	
1715	1220.4709	1263.0000	1291.3691	
1716	1220.4709	1260.0000	1291.3691	
1717	1220.4709	1259.0000	1291.3691	
1718	1220.4709	1240.0000	1291.3691	
1719	1220.4709	1260.0000	1291.3691	
1720	1220.4709	1246.0000	1291.3691	
1721	1220.4709	1238.0000	1291.3691	
1722	1220.4709	1253.0000	1291.3691	
1723	1220.4709	1249.0000	1291.3691	
1724	1220.4709	1245.0000	1291.3691	
1725	1220.4709	1251.0000	1291.3691	
1726	1220.4709	1252.0000	1291.3691	2
1727	1220.4709	1249.0000	1291.3691	
1728	1220.4709	1274.0000	1291.3691	
1729	1220.4709	1258.0000	1291.3691	
1730	1220.4709	1268.0000	1291.3691	
1731	1220.4709	1248.0000	1291.3691	
1732	1220.4709	1295.0000	1291.3691	1
1733	1220.4709	1243.0000	1291.3691	
1734	1220.4709	1253.0000	1291.3691	
1735	1220.4709	1258.0000	1291.3691	

Individual Measurements Chart Summary for weight

id	--3 Sigma Limits with n=2 for Moving Range- Lower Limit	Moving Range	Upper Limit
1711	0	.	43.553759
1712	0	12.000000	43.553759
1713	0	10.000000	43.553759
1714	0	12.000000	43.553759
1715	0	3.000000	43.553759
1716	0	3.000000	43.553759
1717	0	1.000000	43.553759
1718	0	19.000000	43.553759
1719	0	20.000000	43.553759
1720	0	14.000000	43.553759
1721	0	8.000000	43.553759
1722	0	15.000000	43.553759
1723	0	4.000000	43.553759
1724	0	4.000000	43.553759
1725	0	6.000000	43.553759
1726	0	1.000000	43.553759
1727	0	3.000000	43.553759
1728	0	25.000000	43.553759
1729	0	16.000000	43.553759
1730	0	10.000000	43.553759
1731	0	20.000000	43.553759
1732	0	47.000000	43.553759
1733	0	52.000000	43.553759
1734	0	10.000000	43.553759
1735	0	5.000000	43.553759

Output 34.1.2. Tests for Special Causes



Output 34.1.1 and Output 34.1.2 indicate that Test 1 was positive for engine 1732 and Test 2 was positive for engine 1726. Test 1 detects one point beyond Zone A (outside the control limits) and Test 2 detects seven points (TEST2RUN=7) in a row on one side of the central line.

Example 34.2. Specifying Standard Values for the Process Mean and Standard Deviation

See SHWIREX2 in the SAS/QC Sample Library

By default, the IRCHART statement estimates the process mean (μ) and standard deviation (σ) from the data, as in the previous example. However, there are applications in which known (standard) values μ_0 and σ_0 are available for these parameters based on previous experience or extensive sampling.

For example, suppose that the manufacturing process described in the previous example produces engines whose weights are normally distributed with a mean of 1250 and a standard deviation of 12. The following statements create individual measurements and moving range charts based on these values:

```
title 'Specifying Standard Process Mean and Standard Deviation';
symbol v=dot;
proc shewhart data=engines;
   irchart weight*id /
       mu0    = 1250
       sigma0 = 12
       xsymbol= mu0;
run;
```

The charts are shown in Output 34.2.1. The MU0= option and SIGMA0= option specify μ_0 and σ_0. The XSYMBOL= option specifies the label for the central line on the individual measurements chart, and the keyword MU0 requests a label indicating that the central line is based on a standard value.

Output 34.2.1. Specifying Standard Values with MU0= and SIGMA0=



You can also specify μ_0 and σ_0 as the values of the variables _MEAN_ and _STDDEV_ in a LIMITS= data set. For example, the following statements create a LIMITS= data set with the standard values specified in the preceding IRCHART statement:

```
data englim;
   length _var_ _subgrp_ _type_ $8;
   _var_    = 'weight';
   _subgrp_ = 'id';
   _limitn_ = 2;
   _type_   = 'STANDARD';
   _mean_   = 1250;
   _stddev_ = 12;
run;
```

The variables _VAR_ and _SUBGRP_ are required, and their values must match the *process* and *subgroup-variable*, respectively, specified in the IRCHART statement. The bookkeeping variable _TYPE_ is not required, but it is recommended to indicate that the variables _MEAN_ and _STDDEV_ provide standard values rather than estimated values. See "LIMITS= Data Set" on page 1169 for details.

The following statements read ENGLIM as a LIMITS= data set:

```
proc shewhart data=engines limits=englim;
   irchart weight*id / xsymbol=mu0;
run;
```

The resulting charts (not shown here) are identical to those shown in Output 34.2.1.

Example 34.3. Displaying Distributional Plots in the Margin

See SHWIREX3 in the SAS/QC Sample Library

You can augment a chart for individual measurements with one of several graphical displays, such as a histogram or a box-and-whisker plot. These displays summarize the measurements plotted on the chart, and, if the process is in statistical control, they provide a view of the process distribution.

For example, the following statements create an individual measurements chart for the engine weight measurements in the data set ENGINES (see page 1175) augmented with a histogram of the weights:

```
title 'Individual Measurements Chart and Histogram';
symbol v=dot;
proc shewhart data=engines;
   irchart weight*id /
       rtmplot=histogram
       nochart2;
run;
```

The chart is shown in Output 34.3.1. The RTMPLOT= option requests a histogram in the right margin. The NOCHART2 option suppresses the display of the moving range chart.

Output 34.3.1. Histogram in Right Margin



The following *keywords*, requesting different types of plots, are available with the RTMPLOT= option:

Keyword	Marginal Plot
HISTOGRAM	histogram
DIGIDOT	digidot plot
SKELETAL	skeletal box-and-whisker plot
SCHEMATIC	schematic box-and-whisker plot
SCHEMATICID	schematic box-and-whisker plot with outliers labeled
SCHEMATICIDFAR	schematic box-and-whisker plot with far outliers labeled

See the entry for the BOXSTYLE= option in Chapter 46, "Dictionary of Options," for a description of the various box-and-whisker plots.

You can also use the LTMPLOT= option to request univariate plots in the left margin. The following statements request an individual measurements chart with a box-and-whisker plot in the left margin:

```
title 'Individual Measurements Chart and Box-and-Whisker Plot';
symbol v=dot;
proc shewhart data=engines;
   irchart weight*id /  ltmplot =schematic
                        ltmargin=8
                        nochart2;
run;
```

The chart is shown in Output 34.3.2. The same *keywords* that are available with the RTMPLOT= option can be specified with the LTMPLOT= option. The LTMARGIN= option specifies the width (in horizontal percent screen units) of the left margin.

Output 34.3.2. Box-and-Whisker Plot in Left Margin



The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/QC® User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. 1994 pp.

SAS/QC® User's Guide, Version 8

ISBN 1–58025–493–4

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999